Maple Commands for Surfaces and Partial Derivatives

We've seen that a curve can be written as $y = f(x)$, or more generally in parametric form using one parameter (usually t), and the curve can be in two or three dimensions:

$$\mathbf{r}(t) = \langle x(t), y(t) \rangle \qquad \mathbf{r}(t) = \langle x(t), y(t), z(t) \rangle$$

Similarly, a surface can be represented in simple terms as the graph of $z = f(x, y)$, or in parametric form using *two* parameters, for example, u, v:

$$\mathbf{F}(u, v) = \langle x(u, v), y(u, v), z(u, v) \rangle$$

Before we go too far into this, let's review the ways we can plot curves (one dimensional functions) in Maple.

Examples

- Plot the curve $y = x^2$ as the graph of a function:

  ```
  plot(x^2,x=-2..2);
  ```

- Plot the curve $y = x^2$ and $y = \sin(x)$ for $0 \leq x \leq 5$:

  ```
  plot([x^2,sin(x)],x=0..5);  #Using square brackets- Better
  plot({x^2,sin(x)},x=0..5);  #A valid command
  ```

  If you try to put in a legend, you'll get an error in one of these- (Which one?)

- Plot the inverse of the sine function:

  ```
  plot([sin(x),x,x=0..5]); # Parametrized form-
  plot(arcsin(x),x=-1..1); # Direct form
  ```

- Plot the curve $x = \cos(t)$, $y = \sin(3t)$ as a parametric curve, for $0 \leq t \leq 2\pi$:

  ```
  plot([cos(t),sin(3*t),t=0..2*Pi]);
  ```

- Plot the last parametric curve together with the curve $y = 2\cos(x)$:

  ```
  plot([[cos(t),sin(3*t),t=0..2*Pi],[t,2*cos(t),t=0..2*Pi]]);
  ```

Summary

There are several ways of defining a function of 1 variable. Maple has a slightly different way of plotting it based on how the function is defined:

1. The function is $y = f(x)$:

   `plot(f(x),x=a..b)` or in parametric form: `plot([x,f(x),x=a..b]);`

2. The function is $x = f(t), y = g(t)$: `plot([f(t),g(t),t=a..b]);`

3. We haven't discussed this next one, but sometimes it is useful as well: $f(x, y) = c$ is an implicitly defined function. Here is an example, where we plot the implicit function $\cos(xy) = 1/3$:

   ```
   with(plots);
   implicitplot(cos(x*y)=1/3,x=-Pi..Pi,y=-Pi..Pi);
   ```

   We'll use these in the next section.

New Examples

If we want to plot the *surface* $z = f(x, y)$, here are some example commands.

- Plot the upper hemisphere of a sphere of radius 1.

  The equation is $x^2 + y^2 + z^2 = 1$, we can solve for z and take the positive root:

  $$z = \sqrt{1 - x^2 - y^2}$$

  In the plot command, there are several extra options- See if you can figure out what they do (that is, `axes=normal` (or you can use `axes=boxed` and the `scaling=constrained` is by default "unconstrained"). The line below is broken for readability- Don't break the line in Maple.

  ```
  plot3d(sqrt(1-x^2-y^2),x=-1.5..1.5,y=-1.5..1.5,
                        axes=NORMAL, scaling=CONSTRAINED);
  ```

- Plot the surface $z = xe^{-x^2-y^2}$, and color the graph according to $\sin(xy)$:

  ```
  plot3d(x*exp(-x^2-y^2), x = -2..2, y =-2..2, color = sin(x*y))
  ```

  Just for fun, we can also do an animation, where we vary the height of the valley and hilltop by using an extra parameter A:

```
with(plots):
animate( plot3d, [sin(A)*x*exp(-x^2-y^2), x=-2..2, y=-2..2],
                    A=0..2*Pi,shading=zhue );
```

Then click on the graph and there will be a play button on the menu. There is also an option to run a loop- See if you can find it.

- Just as before, the surface may be defined implicitly. For example, the sphere is $x^2 + y^2 + z^2 = 1$, and we might plot it directly:

  ```
  implicitplot3d(x^2+y^2+z^2=1,x=-2..2,y=-2..2,z=-2..2);
  ```

  It looks more like a soccer ball than a sphere- You can get a better plot by inserting the option: `numpoints=1500`

- Plot multiple surfaces:

  Method 1: Plot the two separately, then "display" the results. Use this technique if you want to overlay graphs over the top of each other (or with each other).

  In this example, plot $z = \sin(x)\sin(y)$, then plot the parametrized half sphere (as before).

  ```
  with(plots):
  A:=plot3d(sin(x)*sin(y),x=-Pi..Pi,y=-Pi..Pi,color=white):
  B:=plot3d(x*exp(-x^2-y^2), x = -2..2, y =-2..2, color = sin(x*y));
  display({A,B});
  ```

  Method 2: If your functions are all of the form $z = f(x, y)$, you can use a single plot command. For example, here we plot three surfaces: $\sin(xy)$, $x - y$ and $4\mathrm{e}^{-x^2-y^2}$, all on the same graph:

  ```
  plot3d({sin(x*y), x-y, 4*exp(-x^2-y^2)},x=-2..2,y=-2..2);
  ```

  Question: What happens if you use square brackets around the functions instead of curly braces?... Does the same thing happen with 4 functions?

Level Curves

Also of interest are the level curves of a function. Maple uses the `contourplot` or `contourplot3d` function. For example, we plot the contours for $z = \sqrt{x^2 - y^2}$ for contours $z = 1/2, 3, 4$ by giving the following command (assumes you have already given the `with(plots)` command).

```
contourplot(sqrt(x^2-y^2),x=-5..5,y=-5..5,
                      contours=[1/2,3,4],grid=[100,100]);
```

As a second example, here we use the three-dimensional version of the contour plot, and we use an extra option that helps with the visualization. This surface is called a "dog saddle" (a dog has 4 legs, usually):

```
contourplot3d(x*y*(x^2-y^2),x=-3..3,y=-3..3,filledregions = true);
```

Multivariate Limits

Multivariate limits can be difficult to compute by hand- Maple can have a difficult time as well. However, often Maple is able to compute a multivariate limit- and we use the `limit` command we learned earlier. There are some excellent examples in the help file: `?limit`

Examples

1. Find $\lim_{(x,y)\to(0,0)} \frac{3x^2y}{x^2+y^2}$, if it exists. You should look at the plot first to see if you can first guess.

   SOLUTION in Maple:

   ```
   F:=3*x^2*y/(x^2+y^2);
   plot3d(F,x=-1..1,y=-1..1)
   ```

   We see a nice surface- It may have some edges, but nothing that should give a discontinuity. See if Maple can find the limit:

   ```
   limit(F,{x=0,y=0});
   ```

   Maple's having a hard time here... Using some algebra, we see that:

   $$x^2 \le x^2 + y^2 \quad \Rightarrow \quad \frac{x^2}{x^2+y^2} \le 1 \quad \Rightarrow \quad \frac{3x^2y}{x^2+y^2} \le 3|y|$$

   And with a similar argument, the function is bounded below by $-3|y|$, which we can verify with a plot:

   ```
   plot3d({-3*abs(y),F,3*abs(y),x=-1..1,y=-3..3);
   ```

   Therefore, by the Squeeze Theorem, the limit is 0.

2. Show that the following limit either exists or show that it does not exist:

   $$\lim_{(x,y)\to(0,0)} \frac{y^2\sin^2(x)}{x^4+y^4}$$

   SOLUTION: First plot it to see what it looks like:

```
G:=y^2*(sin(x))^2/(x^4+y^4);
plot3d(G,x=-1..1,y=-1..1,shading=zhue,axes=boxed);
```

If you go to the origin along a purple "valley", the limit will be zero. This would be, for example, along either the $x-$ or $y-$axis:

$$\lim_{x\to 0, y} \frac{y^2 \sin^2(x)}{x^4 + y^4} = 0$$

However, trying to go to the origin along the orange "hill", $y = x$, looks like a different limit. Let's try it: Along $y = x$, the expression becomes

$$\frac{x^2 \sin^2(x)}{2x^4} = \frac{\sin^2(x)}{2x^2} = \frac{1}{2}\left(\frac{\sin(x)}{x}\right)^2$$

You may recall that the limit in the parentheses is 1 (l'Hospital's rule, for example), so that the overall limit is now 1/2- which agrees with our graph.

Therefore, the limit at the origin does not exist, and this function is not continuous at the origin.

This is typical of this kind of discontinuity.

3. Does this limit exist?

$$\lim_{(x,y)\to(0,0)} \frac{x^4 - 4y^2}{x^2 + 2y^2}$$

SOLUTION: Plot in Maple, and you'll see something similar to the last example- This limit does not exist.

```
plot3d((x^4-4*y^2)/(x^2+2*y^2),x=-1..1,y=-1..1,shading=zhue,axes=boxed);
```

From the graph, we see that if we travel along the $y-$axis, the limit is -2, but if we travel along the $x-$axis, the limit is 0 (we would want to actually show the algebra as well).

Derivatives and Partial Derivatives

Partial derivatives are easy in Maple, as are higher derivatives.

- **EXAMPLE:** Given $f(x, y) = \sqrt{x^2 + y^2}$, find f_x, f_{yx} and f_{yyx}

  SOLUTION: (Note that we're differentiating expressions and not functions)

```
f:=sqrt(x^2+y^2);
fx:=diff(f,x);
fyx:=diff(f,y,x);
fyyx:=diff(f,y,y,x);
```

Is f_{yyx} the same as f_{yxy}? Check in Maple.

- Suppose $u(x, y) = \mathrm{e}^x \sin(y)$. Show that u satisfies "Laplace's Equation":

$$u_x x(x, y) + u_{yy}(x, y) = 0$$

SOLUTION: In Maple, this is easy to check:

```
u:=exp(x)*sin(y);
uxx:=diff(u,x,x);
uyy:=diff(u,y,y);
uxx-uyy;
```